Paper 444-2013

Examining Mediator and indirect effects of Loneliness in Social Support on Social Well-Being Using Baron and Kenny and a Bootstrapping Method

Abbas S. Tavakoli, DrPH, MPH, ME; Sue Heiney PhD, University of South Carolina, College of Nursing

Abstract
This study presentation examines mediator effect and indirect effect of loneliness in social support on social well being by using two methods: Baron & Kenny, and Bootstrapping. A cross-sectional data were used here from the longitudinal study randomized trial design in which 185 participants were assigned to the therapeutic group (n=93) who received by teleconference with participants interacting in real time with each other and control group (n=92) who received usual psychosocial care (any support used by the patient in the course of cancer treatment. Baron and Kenny (1986)steps and Hayes (2004) we reused to examine mediation effect. Results of Baron indicated that the relationship between social support and social well being was significant (c =.634 (total effect) (p=.0001)) and that there was significant relationship between mediator and predictor variable (**α** =-2.11 (p=.0001)). Also, previously found significant relationship between social support and social well being w (c' =.595 (direct effect) (p=.0001) when both social support and loneliness were in the model. The indirect effect was .039 and Sobel test was significant (P=.028). Therefore, there was no mediator effect for loneliness in the relationship between social support and social well being. The results of bootstrapping methods indicated the direct effect wares .591 (95% CI: .589-.593 for normal theory and .481- .690 for percentile) and indirect effect was .040 (95% CI: .039-.040 for normal theory and .006-.087for percentile). The result showed both methods had significant indirect effect.

Keywords: SAS, Mediator, Hayes, Baron.

The project described was supported by Award Number R01CA107305 from the National Cancer Institute. The content is solely the responsibility of the authors and does not necessarily represent the official views of the National Cancer Institute or the National Institutes of Health.

At the time a portion of this research was conducted Heiney was employed as Manager, Psychosocial Oncology at Palmetto Health Cancer Centers (PHCC) in Columbia, SC.

Introduction
Mediation is commonly tested in the health sciences. There is often confusion among why the relationship between predictor variable (X) and criterion variable (Y) become non significant when we introduce the third variable in the model. The reason could be the third variable functions as a mediator. Baron & Kenny (1986) define a mediator as a variable to the extent that it accounts for the relation between the independent variable and the outcome variable. Two possibilities of mediator effect are diagrammed below:



The general model is described in terms of mediated effects. We assume multivariate normal distributions and normally distributed error terms throughout. The effect of adding a third variable can be calculated in two ways based on either the difference between two regression parameters (c - c') (indirect effect) or the multiplication of two regression parameters (**α** β). **In the first method**, the following two regression equations are estimated.

Model 1: $Y = i_1 + cX + \varepsilon_1$
Model 2: $Y = i_2 + c' X + \beta M + \varepsilon_2$
Model 3: $M = i_3 + \alpha X + \varepsilon_3$

Where Y is the outcome variable, X is independent variable, M is the mediator, c (total effect) codes the relationship between the independent to the outcome in the first equation, c' (direct effect) is the coefficient relating the independent to the outcome adjusted for the effects of the mediator, $\boldsymbol{\varepsilon}_1$ and $\boldsymbol{\varepsilon}_2$ code unexplained variability, and the intercepts are a_1 and a_2.**A second method** also involves estimation of two regression equations (model2 & 3). The indirect t effect or mediator effect is calculated by multiply β coefficient from model 2 and **α** coefficient from model 3. Baron and Kenny (1986) have discussed four steps in establishing mediation:

Step 1: Show that the initial variable is correlated with the outcome (**Model Y =X).**
Step 2: Show that the initial variable is correlated with the mediator **(Model M = X)**.
Step 3: Show that the mediator affects the outcome variable **(Model Y = M X)**.
Step 4: To establish that M completely mediates the X-Y relationship, the effect of X (IV) on Y (DV) controlling for M should be zero (estimate and test path c'). The effects in both Steps 3 and 4 are estimated in the same regression equation.
If all four of these steps are met, then the data are consistent with the hypothesis that variable M **completely mediates** the X-Y relationship, and if the first three steps are met but the Step 4 is not, then **partial mediation** is indicated.

Sobel's (1982) test of significance is performed to determine the extent to which a mediator contributed to the total effect on the outcome variable. Sobel test assume normality and no measurement error. Aroian (1994) and Goodman (1960) proposed different standard error to test indirect effect. Hayes (2010) method estimates and tests the indirect effect of independent variable (X) on dependent viable (Y) through mediation (M) by using bootstrapping. Hayes (2004) used bootstrapping to examine the indirect effect.

Purpose

The purpose of this paper is to examine mediator and indirect effect of loneliness in social support (SSQ) on social well being (SWB) by two methods: Baron & Kenny and bootstrapping.

Background

The cross-sectional data from research grant (R01) used here were collected in the first of three interviews of a longitudinal study designed to test and compare the effects of a therapeutic group using teleconference for African American women with breast cancer on social disconnection, a sense of being cut off from partners, family and friends due to side effects of treatment and fatalistic beliefs about cancer. A therapeutic group by teleconference may assist African American women with breast cancer to feel connected to women in a similar situation, to learn ways to talk about cancer and to decrease fatalistic beliefs. A randomized trial design was used in which 185 participants were assigned to the therapeutic group (n=93) and control group (n=92). The therapeutic group intervention was led by two social workers experienced in working with oncology patients and leading support groups. The intervention was delivered by teleconference with participants interacting in real time with each other. Control group was defined as any support used by the patient in the course of cancer treatment. The randomization was stratified by treatment type. Data were collected at baseline, the end of the intervention (8 weeks from baseline), and 16 weeks from baseline.

Data Analyses

All data analyses were performed using **SAS/STAT** ® statistical software, version 9.3 (SAS, 2008). Proc MEAN used to describe the data. PROC CORR and REG were used to analyze this study. Bootstrapping used to estimate indirect effect through the repeated sampling of data. Also, two different methods (normal distribution theory and percentile) used to calculate confidence interval for indirect effect. Since all variables were continuous standard Pearson correlation and regression procedures were used to examine the interrelationships among the study variables. *P*-values less than or equal to .05 were considered significant.

Results

Table 1 showed descriptive statistics for social well being social support, and loneliness. Table 2 revealed descriptive statistics for variables using bootstrap. The result showed the means from original sample and bootstrapping are identical. However, the standard deviation in bootstrapping is smaller than original sample. Table 3 indicated pairwise

Pearson correlation. The result s revealed positive correlation between social well being and social support (r=.67) and negative linear relationship between social well being and loneliness (r=-.33).
To test for mediation (Baron & Kenny method), three regression equations were run for each purpose. First, the outcome (social well being) was regressed on the predictor variable (social support).This relationship was significant (c =.63 (p=.0001)). Therefore, we ran second and third equations were analyzed. In the second equation, the mediator (loneliness) was regressed on the predictor variable (social support). The result indicated that there was significant relationship between mediator and predictor variable (**α** =-2.11 (p=.006)). The third equation involved regressing the outcome (social well being) variable simultaneously on the predictor (social support) and mediator variable (loneliness). The result indicated that the previously significant relationship between predictor (social support) and the outcome (social well being) remained significant (c' =.59 (p=.0001)). Therefore, there is no evidence of mediator effect for loneliness in the relationship between social support and social well being (see Table 4 & Figure 1). Table 5 indicated three different tests (Sobel, Goodman, and Aroian) to examine the indirect effect for original sample. All tests were statistically significant (P=.0218 to .0285) which indicated there was indirect effect. Table 6 showed result of indirect effect using bootstrap method. The results of bootstrapping method indicated the direct effect wares .591 with 95% CI: .589-.593 for normal theory and .481- .690 for percentile. Also, indirect effect was .040 with 95% CI: .039-.040 for normal theory and .006-.087for percentile.

Conclusion

This paper examined the influence of loneliness in the relationship between social support and social well being, whether the relationship was influenced by a mediator effect. The result revealed that there was not mediator effect for loneliness in the relationship between social support and social well being. However, the result indicated that there was significant indirect effect with both Sobel, Goodman, and Aroian tests and bootstrapping. Those tests for indirect effect are valid when the assumption of normality of the sampling distribution can be met. Bootstrapping are powerful technique to calculate confidence interval for indirect effect without any assumptions about sampling distribution. Therefore, our recommendation is to use bootstrapping to examine indirect effect.

References

1. Baron, R. M., & Kenny, D. A. (1986). The moderator-mediator variable distinction in social psychological research: Conceptual, strategic, and statistical consideration. Journal of Personality & Social Psychology, 51, 1173-1182.
2. Goodman, L. A. (1960). On the exact variance of products. *Journal of the American Statistical Association*, **55**, 708-713.
3. Hayes, A. F., Preacher, K.J. (2010). Estimating and testing indirect effects in simple mediation models when the constituent paths are nonlinear. Multivariate Behavioral Research, 45, 627-660.
4. MacKinnon, D.P. & Dwyer, J.H. (1993). Estimating mediated effects in prevention studies. *Evaluation Review*, 17, 144-158
5. Sobel, M. E. (1982). Asymptotic confidence intervals for indirect effects in structural equation models. Sociological Methodology **13**: 290–312

Contact Information

Abbas S. Tavakoli, DrPH, MPH, ME
College of Nursing
University of South Carolina
1601 Greene Street
Columbia, SC 29208-4001
Fax: (803) 777-5561
E-mail: abbas.tavakoli@sc.edu

Table 1. Measure of center and dispersion for variables

Variable*	N	Mean	Std Dev	Minimum	Maximum
SWB	185	23.66	4.86	5.83	28.00
SSQ	185	34.77	5.15	12.00	40.00
Worry	185	43.11	53.92	0.00	200.00

*. SWB (Social Well Being), SSQ (Social Support), & Worry (Loneliness).

Table 2. Measure of center and dispersion for variables using bootstrap

Variable*	N	Mean	Std Dev	Minimum	Maximum
mswb	5000	23.66	0.35	22.27	24.79
mssqf	5000	34.77	0.38	33.35	36.05
mworry	5000	43.12	3.92	31.08	57.38

*. mswb (Social Well Being), mssqf (Social Support), & mworry (Loneliness).

Table 3. Pairwise Pearson correlation

Pearson Correlation Coefficients, N = 185 Prob > |r| under H0: Rho=0			
	SWB	**SSQ**	**Worry**
SWB	1.00000	0.67081 <.0001	-0.33130 <.0001
SSQ		1.00000	-0.20112 0.0060
Worry			1.00000

*. SWB (Social Well Being), SSQ (Social Support), & Worry (Loneliness).

Table 4. Regression results Baron & Kenny method

Model : SWB = SSQ (Y=X)

Analysis of Variance					
Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	1	1958.18225	1958.18225	149.72	<.0001
Error	183	2393.51084	13.07929		
Corrected Total	184	4351.69309			

Root MSE	3.61653	**R-Square**	0.4500
Dependent Mean	23.66396	**Adj R-Sq**	0.4470
Coeff Var	15.28286		

Parameter Estimates								
Variable*	**DF**	**Parameter Estimate**	**Standard Error**	**t Value**	**Pr > |t|**	**Standardized Estimate**	**Squared Semi-partial Corr Type II**	**Squared Partial Corr Type II**
Intercept	1	1.62443	1.82074	0.89	0.3735	0	.	.
SSQ	1	0.63391	0.05181	12.24	<.0001	0.67081	0.44998	0.44998

Model: Worry = SSQ (M=X)

Parameter Estimates								
Variable*	**DF**	**Parameter Estimate**	**Standard Error**	**t Value**	**Pr > |t|**	**Standardized Estimate**	**Squared Semi-partial Corr Type II**	**Squared Partial Corr Type II**
Intercept	1	116.37000	26.66283	4.36	<.0001	0	.	.
SSQ	1	-2.10719	0.75867	-2.78	0.0060	-0.20112	0.04045	0.04045

Model: SWB = SSQ Worry (Y = X M)

Parameter Estimates								
Variable*	**DF**	**Parameter Estimate**	**Standard Error**	**t Value**	**Pr > |t|**	**Standardized Estimate**	**Squared Semi-partial Corr Type II**	**Squared Partial Corr Type II**
Intercept	1	3.77259	1.84699	2.04	0.0425	0	.	.
SSQ	1	0.59501	0.05106	11.65	<.0001	0.62964	0.38042	0.42732
Worry	1	-0.01846	0.00487	-3.79	0.0002	-0.20466	0.04019	0.07307

*. SWB (Social Well Being), SSQ (Social Support), & Worry (Loneliness).

<u>Figure 1 Mediator Model:</u> Loneliness (Worry) as mediator of social support (SSQ) to social well being (SWB)

<u>Step. 1</u>

β =.633 (p=.0001)

Social Support ------------------- --------→ Social Well being

<u>Step 2 and 3.</u>

Figure1: Loneliness (Worry) as Mediator of Social Support (SSQ to Social Well Being (SWB).



Indirect Effect= c – c' = .633 – (.595) = .038

Table 5. Indirect test results

a:SSQ(X)Coeff Model M=x	sa:Standard error (X) Model M=x	b:Worry(M) Coeff Model Y=X M	sb:SE(M) Model y = X M	Total Effect
-2.10719	0.75867	-0.018460	.004873373	0.63391

Standarderror total effect	ratio indirect effectover total effect	ratio indirect effect over direct effect	Sobel test	P value Sobel test
0.051808	0.061362	0.065374	2.23986	0.025100

Goodman I test	Pvalue Goodman test	Goodman II test	Pvalue Aroiantest
2.19076	0.028469	2.29242	0.021882

Table 6. Results of Bootstrap (n=5000)

Method I: Normal Distribution CI

Means ssq Bootstrap direct effect	Standarderror ssq	lowerCI ssq	upperCI ssq	Means indirecteffect Bootstrap	Standarderror indirect effect
0.59129	.000744922	0.58983	0.59275	0.039648	.000292914

lowerCI indirect effect	upperCI indirect effect	Means total effect Bootstrap	Standard error totaleffect Bootstrap	Lower CI total effect	Upper CI total effect
0.039074	0.040223	0.63094	.000798095	0.62937	0.63250

Method 2: Percentile CI

Direct effect ssq Original	Direct effect Means ssq Bootstrap	Direct effect lower CI ssq	Direct effect upper CI ssq	worry Original	Means worry Bootstrap	Lower CI worry	Upper CI worry
0.59501	0.59129	0.48058	0.69034	-0.018460	-0.018596	-0.029559	-.008386395

Indirect effect Original	Means indirect effect Bootstrap	lower CI indirect effect	upperCI indirect effect	totaleffect Original	Means totaleffect Bootstrap	lowerCI total effect	upperCI total effect
0.038898	0.039648	.005954625	0.086787	0.63391	0.63094	0.51310	0.73938

Attachment

SAS Syntax

```
**Let us first get a new file with 5000 ;
%let rep=5000;
 proc surveyselect data=path out=outboot
seed=3292012   method=urs  samprate=1
outhits rep=&rep; run;
**** Descriptive and correlation ***;
ods rtf; ods listing close;
proc means data=path maxdec=2;
     var    swb ssq_family worry ;
     title ' means '; run;

proc corr data=path;
     var    swb ssq_family worry ;
     title ' Correlation '; run;

proc means data=outboot noprint ;
     var  swb ssq_family worry ;
    by replicate;
    output out=outall mean= mswb mssqf mworry;    run;

proc means data=outall maxdec=2;
    var  mswb  mssqf  mworry; run;
    ods rtf close; ods listing; quit; run;

ods rtf; ods listing close;
**** Baron & Kenny method ***;
%macro regb  (d,i,t);
proc reg data=path  ;
      model &d = &i / stb pcorr2 scorr2;
      title ' Regression model/ ' &t;
       title3 ' Baron Kenny method';
     %mend regb;
%regb (swb,ssq_family , social support Step 1: Y=x );
%regb (worry,ssq_family , social support Step 2: m=x );
%regb (swb,ssq_family worry , social support Step 3: Y=x m );
run;
ods rtf close; ods listing; quit; run;

*** Soble , Goodman, Aroian tests ****;
data test;
            set path; x= ssq_family ; m= worry ; y= swb ;  run;

proc reg data=test noprint;
    model y=x;    model m=x;     model y=x m;
   ods output ParameterEstimates=regout;     run; quit;

* Here we select only the statistics required to compute the tests *;
 data stest;
```

```
        set regout;
        if model = 'MODEL2' and variable = 'x' then a  = Estimate;
        if model = 'MODEL2' and variable = 'x' then sa = StdErr;
        if model = 'MODEL3' and variable = 'm' then b = Estimate;
        if model = 'MODEL3' and variable = 'm' then sb = StdErr;
        if model = 'MODEL1' and variable = 'x' then te = Estimate;
  if model = 'MODEL1' and variable = 'x' then se = StdErr; run;
proc summary data=stest nway;
   var a sa b sb te se; output out=stesto max= ; run;

proc format;
   value Testf  1 = 'Sobel'  2 = 'Goodman'  3 = 'Aroian';

data stest2; set stesto;
zvs = (a*b)/sqrt(((b*b)*(sa*sa))+((a*a)*(sb*sb)));
abssobel = abs(zvs);
ps = 2*(1-CDF('NORMAL',abssobel));
toteff = (a*b)/((a*b)+(te-(a*b)));
ratio = (a*b)/((te-(a*b)));
Test=1;
format test testf.; run;

*******************************;
* Goodman test *;
*******************************;
data stest3; set stesto;
zvgi = (a*b)/sqrt(((b*b)*(sa*sa))+((a*a)*(sb*sb))+((sa*sa)*(sb*sb)));
absgood = abs(zvgi);
pgi = 2*(1-CDF('NORMAL',absgood));
Test=2;
format test      testf.; run;

*******************************;
* Aroian test *;
*******************************;
data stest4; set stesto;
zvgii = (a*b)/sqrt(((b*b)*(sa*sa))+((a*a)*(sb*sb))-((sa*sa)*(sb*sb)));
absgood2 = abs(zvgii);
pgii = 2*(1-CDF('NORMAL',absgood2));
Test=3;
format test testf.; run;

data stest5 (drop = _type_ _freq_ test);
Merge stest2 stest3 stest4; run;

ods rtf; ods listing close;
proc print split = '*' data=stest5 ;
 var a sa b sb te se  toteff ratio abssobel ps absgood pgi absgood2 pgii;
label
  a  ='a: SSQ(X)Coeff*Model M=x'
  sa = 'sa: Standard error (X)* Model M=x'
```

```
   b = 'b: Worry (M) Coeff* Model  Y=X M'
   sb = 'sb: SE (M)* Model y = X M'
   te = 'Total Effect'
   se = ' Standard error* total effect'
   toteff = ' ratio indirect effect over * total effect'
   ratio  = ' ratio indirect effect over * direct effect'
   abssobel = ' Sobel test'
   ps     = ' P value * Sobel test'
   absgood        = ' Goodman test'
   pgi  = 'P value * Goodman test'
   absgood2 = ' Aroian test'
   pgii = ' P value * Aroian test' ;
   Title ' Indriect tests'; run;
ods rtf close; ods listing;  quit; run;

*** Bootstarpping method  ***;
%macro regcf  (d,i,t);

proc reg data=outboot outest=fst3 noprint;
      by  replicate;
      model &d = &i / stb pcorr2 scorr2;
      title ' Regression model/ ' &t;

%mend regcf;
 %regcf (swb,ssq_family worry, social support and loneliness on well being);
run; quit;

%macro regcs  (d,i,t);
proc reg data=outboot outest=sst3 noprint;
      by  replicate;
      model &d = &i / stb pcorr2 scorr2;
      title ' Regression model/ ' &t;
        %mend regcs;
%regcs (worry,ssq_family , social support  m=x );
run; quit;

data fst3 (drop= ssq_family worry swb);   set fst3;
fssqf = ssq_family;
fworry = worry; run;

data sst3 (drop= ssq_family worry);   set sst3;
sssqf = ssq_family; run;

data ast3;    merge fst3 sst3; run;

data st3 (drop =  swb );    set ast3;
inef = sssqf*fworry;
teff = fssqf + inef; run;
proc means data= st3 maxdec=2 noprint ;
     var fssqf fworry inef teff;
     output out=st3out      mean = mssqf mworry minef mteff
```

```
                stderr=sssqf sworry sinef steff
                n = nssqf nworry ninef nteff
                lclm = lssqf lworry linef lteff
                uclm = ussqf uworry uinef uteff; run;

data st3out (drop = _type_ _freq_);     set st3out; run;

data all ;
  merge st st3out ;
bssqf = ssqfo - mssqf;
bworry = worryo - mworry;
binef = inefo - minef; Run;

*** Method I to calculate % cI for effects  ***;
lssqf = mssqf - (tinv (.95, nssqf-1)*sssqf);
ussqf = mssqf + (tinv (.95, nssqf-1)*sssqf);
lworry = mworry - (tinv (.95, nworry-1)*sworry);
uworry = mworry + (tinv (.95, nworry-1)*sworry);
linef = minef - (tinv (.95, ninef-1)*sinef);
uinef = minef + (tinv (.95, ninef-1)*sinef);
lteff = mteff  - (tinv (.95, nteff -1)*steff );
uteff = mteff  + (tinv (.95, nteff -1)*steff);
run;

ods rtf; ods listing close;
proc print  split = '*' data = all ;
  var  mssqf sssqf lssqf ussqf
       minef sinef linef uinef
       mteff steff lteff uteff;
label
      mssqf = 'Means*ssqfamily*Bootstrap*direct effect'
      sssqf = 'Standard error*ssq'
      lssqf = 'lower CI*ssq'
      ussqf = 'upper CI* ssq'
      minef = 'Means*indirect effect*Bootstrap'
      sinef = 'Standard error*indirect effect'
      linef = 'lower CI*indirect effect'
      uinef = 'upper CI*indirect effect'
      mteff = 'Means*total effect*Bootstrap'
      steff = 'Standard error*total effect*Bootstrap'
      lteff = 'lower CI*total effect'
      uteff = 'upper CI*total effect';

     title 'printing result  ';
      title3 ' 95% CI / Method I  ';  run;
ods rtf close; ods listing; quit;  run;
*** MEthod II to calculate % cI for effects  ***;
%LET ALPHA=.05;
%let a1 = %sysevalf (&alpha/2*100);
%let a2 = %sysevalf ((1-&alpha/2)*100);
```

```
proc univariate data =st3 alpha=.05;
      var fssqf fworry inef teff;

output out=pmeth mean = mfssqf mworry minef mteff
                 pctlpts =&a1 &a2
                 pctlpre = lssqf lworry linef lteff
                 pctlname= p025 p975; run;
            title 'univariate  result  ';
            title3 ' 95% CI / Method  II '; run;

data allu ;     merge st pmeth ;
lssqfu = lssqfp025;
lworryu = lworryp025;
linefu  = linefp025;
lteffu  = lteffp025;
ussqfu = lssqfp975;
uworryu = lworryp975;
uinefu  = linefp975;
uteffu  = lteffp975;
bssqfu = ssqfo - mfssqf;
bworryu = worryo - mworry;
binefu = inefo - minef;
bteffu = teffo - mteff;  run;
ods rtf; ods listing close;

proc print  split = '*' data = allu ;
   var ssqfo mfssqf    lssqfu ussqfu           ;
   var    worryo mworry   lworryu uworryu;
    var inefo minef   linefu uinefu ;
    var teffo mteff   lteffu uteffu;

label
 ssqfo = 'Direct effect*ssq*Original'
       mfssqf = 'Direct effect *Means*ssq*Bootstrap'
       lssqfu = 'direct effect*lower CI*ssq'
       ussqfu = 'direct effect *upper CI* ssq'
       worryo = 'worry*Orginal'
       mworry = 'Means*worry*Bootstrap'
       lworryu = 'lower CI*worry'
       uworryu = 'upper CI*worry'
       inefo = 'indirect effect*Original'
       minef = 'Means*indirect effect*Bootstrap'
       linefu = 'lower CI*indirect effect'
       uinefu = 'upper CI*indirect effect'
       teffo = 'total effect*Orginal'
       mteff = 'Means*total effect*Bootstrap'
       lteffu = 'lower CI*total effect'
       uteffu = 'upper CI*total effect '       ;
      title 'printing result/ second method  ';
      title3 ' 95% CI / Method  II ';run;
ods rtf close; ods listing;  quit; run;
```